Training Guide

TM-1880

AVEVA Everything3D

PML Macros and Functions

Copia para EE AA



Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Revision Log

Date	Revision	Description	Author	Reviewed	Approved

Updates

Change highlighting will be employed for all revisions. Where new or changed information is presented section headings will be highlighted in Yellow.

Suggestion / Problems

If you have a suggestion about this manual or the system to which it refers please report it to AVEVA Training & Product Support at **tps@aveva.com**

This manual provides documentation relating to products to which you may not have access or which may not be licensed to you. For further information on which products are licensed to you please refer to your licence conditions.

Visit our website at **http://www.aveva.com**

Disclaimer

1.1 AVEVA does not warrant that the use of the AVEVA software will be uninterrupted, error-free or free from viruses.

1.2 AVEVA shall not be liable for: loss of profits; loss of business; depletion of goodwill and/or similar losses; loss of anticipated savings; loss of goods; loss of contract; loss of use; loss or corruption of data or information; any special, indirect, consequential or pure economic loss, costs, damages, charges or expenses which may be suffered by the user, including any loss suffered by the user resulting from the inaccuracy or invalidity of any data created by the AVEVA software, irrespective of whether such losses are suffered directly or indirectly, or arise in contract, tort (including negligence) or otherwise.

1.3 AVEVA's total liability in contract, tort (including negligence), or otherwise, arising in connection with the performance of the AVEVA software shall be limited to 100% of the licence fees paid in the year in which the user's claim is brought.

1.4 Clauses 1.1 to 1.3 shall apply to the fullest extent permissible at law.

1.5 In the event of any conflict between the above clauses and the analogous clauses in the software licence under which the AVEVA software was purchased, the clauses in the software licence shall take precedence.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Copyright

Copyright and all other intellectual property rights in this manual and the associated software, and every part of it (including source code, object code, any data contained in it, the manual and any other documentation supplied with it) belongs to, or is validly licensed by, AVEVA Solutions Limited or its subsidiaries.

All rights are reserved to AVEVA Solutions Limited and its subsidiaries. The information contained in this document is commercially sensitive, and shall not be copied, reproduced, stored in a retrieval system, or transmitted without the prior written permission of AVEVA Solutions Limited. Where such permission is granted, it expressly requires that this copyright notice, and the above disclaimer, is prominently displayed at the beginning of every copy that is made.

The manual and associated documentation may not be adapted, reproduced, or copied, in any material or electronic form, without the prior written permission of AVEVA Solutions Limited. The user may not reverse engineer, decompile, copy, or adapt the software. Neither the whole, nor part of the software described in this publication may be incorporated into any third-party software, product, machine, or system without the prior written permission of AVEVA Solutions Limited, save as permitted by law. Any such unauthorised action is strictly prohibited, and may give rise to civil liabilities and criminal prosecution.

The AVEVA software described in this guide is to be installed and operated strictly in accordance with the terms and conditions of the respective software licences, and in accordance with the relevant User Documentation.

Unauthorised or unlicensed use of the software is strictly prohibited.

Copyright 1974 to current year. AVEVA Solutions Limited and its subsidiaries. All rights reserved. AVEVA shall not be liable for any breach or infringement of a third party's intellectual property rights where such breach results from a user's modification of the AVEVA software or associated documentation.

AVEVA Solutions Limited, High Cross, Madingley Road, Cambridge, CB3 0HB, United Kingdom

Trademark

AVEVA and Tribon are registered trademarks of AVEVA Solutions Limited or its subsidiaries. Unauthorised use of the AVEVA or Tribon trademarks is strictly forbidden.

AVEVA product/software names are trademarks or registered trademarks of AVEVA Solutions Limited or its subsidiaries, registered in the UK, Europe and other countries (worldwide).

The copyright, trademark rights, or other intellectual property rights in any other product or software, its name or logo belongs to its respective owner.

Copia para EE AA

CONTENTS

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Copy para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

1 Introduction

This manual is designed to give an introduction to the AVEVA Programming Macro Language. There is no intention to teach software programming but only provide instruction on how to customise AVEVA Everything3D (E3D) using Programmable Macro Language (PML).

This training guide is supported by the reference manuals available within the products installation folder. References will be made to these manuals throughout the guide.

1.1 Aim

The following points need to be understood by the trainees:

- Understand how PML can be used to customise AVEVA E3D.
- Understand how to create Functions, Forms and Objects.
- Understand how to use the built-in features of AVEVA E3D.
- Understand the use of Add-ins to customise the environment.

1.2 Objectives

At the end of this training, the trainees will have a:

- Broad overview of Programmable Macro Language (PML).
- Basic coding practices and conventions.
- How PML can interact with the Design model.
- How Forms and Menus can be defined with PML.

1.3 Prerequisites

The participants must have completed an AVEVA Basic Design Course and have a familiarity with E3D. Previous experience of programming is of benefit – but is not necessary

1.4 Course Structure

Training will consist of oral and visual presentations, demonstrations, worked examples and set exercises. Each trainee will be provided with some example files to support this guide. Each workstation will have a training project, populated with model objects. This will be used by the trainees to practice their methods, and complete the set exercises.

1.5 Using this Guide

Certain text styles are used to indicate special situations throughout this document, here is a summary;

Menu pull downs and button press actions are indicated by bold dark turquoise text.

Information the user has to key-in will be in **bold red text.**

Additional information will be highlighted

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Reference to other documentation will be separate

System prompts should be bold and italic in inverted commas i.e. ***'Choose function'***

Example files or inputs will be in the `courier new font` with colours and styles used as before.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

2 PML Overview

Programmable Macro Language (PML) is the customisation language used by AVEVA E3D and it provides a mechanism for users to add their own functionality to the AVEVA E3D software family. This functionality could be as simple as a re-naming macro, or as complex a complete user-defined application (and everything in between).

PML is a coding language specific to AVEVA products based on the command syntax that is used to drive E3D. As the product develops, PML is also improved providing new functionality and bringing it closer to other object-orientated programming languages, while still retaining the powerful command syntax. Although it is one language, there are three distinct parts:

- PML 1 the first version of PML based on command syntax. String based and provides IF statement, loops, variables & error handling.
- PML 2 object oriented language extending the ability of PML. Use of functions, objects and methods to process information (also covered in TM-1402 – Form Design).
- PML .NET provides the platform in PML to display and use objects created in other .NET languages (covered in TM-1402 – Form Design)

2.1 PML 1 – String-Based Command Syntax

When PML is written as command syntax, E3D processes it as individual lines of command and runs them in sequence - as if they had been typed directly into the command window.

A simple macro is likely to be written completely in command syntax and allows users to re-run popular commands. Saved as an ASCII file, the macro can be run in E3D through the command window (by typing **$m/FILENAME** or by dragging and dropping the file).

2.1.1 Example of a Simple Command Syntax Macro

The following is an example of a simple macro that can be used to create an Equipment element. As each line is run in order, the same piece of equipment will be created each time the macro is run.

```
NEW EQUIP /ABCD
NEW BOX
XLEN 300 YLEN 400 ZLEN 600
NEW CYL DIA 400 HEI 600
CONN P1 TO P2 OF PREV
```

Macros can be extended to include IF statements, DO loops, variables and error handling (explained further later in the course). If additional information is typed onto the same line as the call for the macro, then these become input parameters and are available for use within the macro.

For example, an existing macro could be called by using the command **$M/BUILDBOX 100 200 300**.

In this example, the three numbers (100, 200 & 300) become the three parameters supplied to the macro.

2.1.2 Examples of Command Syntax

The following are examples of command syntax that can be typed directly into the command window (or used within a macro):

Working with elements and attributes:

- To find out attribute information about the current element, type **Q ATT**
- To create an new element (for example, BOX), type **NEW BOX**

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

- To find out a specific attribute (for example NAME), type **Q NAME**
- To set a value to an attribute on the current element (for example XLEN), type **XLEN 300**

Manipulating the drawlist in DESIGN:

- To add the current element to the drawlist, type **ADD CE**
- To add a specific element below a piece of equipment, type **ADD ONLY /E1301-S1**
- To remove all the elements from the drawlist, type **REM ALL**

Annotating elements in DESIGN:

- To label the current element with its name, type **MARK CE**
- To label the element /PIPE with its name, type **MARK /100-B-1-B1**
- To remove the label from the current element, type **UNMARK CE**
- To remove all labels, type **UNMARK ALL**
- To mark all branch elements with their names, type **MARK WITH (NAME) ALL BRAN**
- To mark all large valve elements with their specification reference, type **MARK WITH (NAME OF SPREF) ALL VALV WHERE CPAR[1] GT 100**

Using Aid graphics:

- To draw an unnumbered graphic aid line, type **AID LINE E0N0U0 TO E0N1000U1000**
- To draw a sphere aid (aid number 1000), type **AID SPHERE NUM 1000 E0N0U0 DIAM 200**
- To remove all unnumbered graphical aid lines, type **AID CLEAR LINE UNN**
- To remove all number 1000 sphere aids, type **AID CLEAR SPHERE 1000**
- To remove all graphical aids, type **AID CLEAR ALL**

Adding colour to the 3D model:

- To apply the standard enhance colour to the current element, type **ENHANCE CE**
- To apply colour 10 to element /PIPE, type **ENHANCE /PIPE COL 10**
- To remove all colour from the 3D model, type **UNENHANCE ALL**

Position and orientation of elements:

- To move the current element 1000mm in a N45E direction, type **MOVE N45E DIST 1000**
- To move the current element E, until it is in line with /BOX, type **MOVE E THRU /BOX**
- To rotate the current element around its origin (pointing up) by 45, type **ROTATE BY 45 ABOUT U**
- To relatively move the current element east by 1000mm, type **BY E 1000**
- To explicitly position the current element, type **AT E0 N1000 U2000**
- To position & orientate the current element based on /EQUIP-N1, type **CONNECT P2 TO P0 OF /EQUIP-N1**

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

- To position & orientate the current element based on the previous, type **`CONNECT P3 TO P1 OF PREV`**

2.1.3 Syntax Graphs

Many examples of command syntax are provided in the various reference manuals within the installation folder of AVEVA E3D. For each a syntax graph is provided to show the various combinations of syntax available:



The above syntax graph is for the ADD syntax (used to add elements to the drawlist). From the graph it can be seen that only two words are required (e.g. ADD /PIPE) but others can be included (e.g. ADD ONLY /PIPE1 /PIPE2 COL 3).

The **$Q** syntax can also tell you the next allowable part of the syntax. For example **ADD $Q-10** would print the 10 next available words that can follow the command **ADD**.

More examples of command syntax and the supporting syntax graphs can be found in the relevant reference manuals provided with AVEVA E3D.

2.2 PML 2 – Object Orientated Programming

PML 2 is almost an object-oriented language and it provides most features of other Object-Orientated Programming (OOP) languages. Operators and methods are polymorphic and overloading of methods is supported. However, there is no inheritance and no concept of public or private variables.

PML 2 provides for classes of built-in, system-defined and user-defined object types. Objects have members (their own objects) and methods (their own functions). All PML variables instances of (1) built-in, (2) system-defined or (3) user-defined objects.

Through the use of method concatenation, it is possible to achieve multi-operations in a single line of code. This means that PML 2 methods are typically shorter and easier to read than the PML 1 equivalent. While most PML 1 macros will still run within E3D, PML 2 brings many new features that were previously unavailable. If used together with command syntax it must be expressed as a string before use.

2.2.1 Features of PML 2

The main features of PML 2 are:

- Available built-in variable types - STRING, REAL, BOOLEAN, ARRAY
- Built in Methods for commonly used actions
- Global Functions supersede old style macros
- User Defined Object Types
- PML Search Path (PMLLIB)
- Dynamic Loading of Forms, Functions and Objects (no need for synonyms)

2.2.2 Examples of Object-Orientated PML

Declaration of objects (variables):

- To declare a local variable as a REAL 3, type **`!realVariable = 3`**
- To declare a global BOOLEAN variable as false, type **`!!booleanVariable = FALSE`**

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

- To set the third value in an ARRAY as Fred, type **!arrayVariable[3] = |Fred|**
- To declare a variable as an empty position object, type **!position = object position()**

Finding out information about objects (variables)

- To find out a variable, type **q var !exampleObject**
- To find out the object type of a variable, type **q var !exampleObject.objectType()**
- To find out what members an object has, type `q var !exampleObject.attributes()`

Methods available on objects:

- To find out what methods are available on an object, type **q var !exampleObject.methods()**
- To find out is an object has been given a value, type **q var !exampleObject.set()**
- To find out how large an ARRAY variable is, type **q var !arrayVariable.size()**
- To find out how long a STRING variable is, type **q var !stringVariable.length()**

Clearing an object

- To empty an object, type **!exampleObject.clear()**
- To delete an object that is no longer needed, type **!exampleObject.delete()**

2.2.3 Software Customisation Reference Manual

Many examples of the major objects available in AVEVA E3D are provided in the Software Customisation Reference Manual within the products installation folder. For each object, the members and methods are listed and explained.

- For each of the members the name, type and purpose is provided.
- For each method the name, argument types, returned object type and purpose are provided.

📖 *Refer to the Software Customisation Reference Manual for more information about the major objects.*

2.3 PML Objects

Every variable defined with PML has an object type. This type is set when the variable is defined and is fixed for the life of the variable. When assigning a value to a variable, the object type needs to be considered, otherwise an error may occur. There are three groups of object types available:

- Built-in (e.g. String, Real, Boolean and Array)
- System-defined (e.g. Position, Orientation)
- User-defined

📖 *Refer to the Software Customisation Reference Manual for more examples of System-defined objects*

When a variable is declared as a specific object type, it is given all the members (attributes) and methods of the object definition. This means standard groupings can be setup to store data and that code repetition can be avoided.

A user-defined object provides an opportunity to group data together for a specific purpose. Once grouped as an object, it can be assigned to a variable and used as any other object. The following are examples of two user-defined objects:

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

```
define object FACTORY
  member .name is STRING
  member .workers is REAL
  member .output is REAL
endobject

define object PRODUCT
  member .productCode is STRING
  member .total is REAL
  member .site is FACTORY
endobject
```

These objects would be defined as two separate object files (.pmlobj) and loaded into E3D. You will notice that the object PRODUCT is able to have a member which is another user-defined object. This means that the PRODUCT object has access to all the members and methods of a FACTORY object.

2.3.1 Creating Variables (instances of objects)

There are two types of variables in PML (1) Local and (2) Global

The difference between the two is that global variables last for the whole E3D session and can be referenced directly from other PML routines. Local variables are only available within the routine which defined them. The variables a declared with a single '!' for local and a double '!!' for global.

- !localVariable – a local variable
- !!globalVariable – a global variable

A variable object-type can be implied from the assigned value:

```
!name = |Fred|        $* To create a LOCAL, STRING variable

!!answer = 42         $* To create a GLOBAL, REAL variable

!!flag = TRUE         $* To create a GLOBAL, BOOLEAN variable
```

It is also possible to define a variable as an object-type without an initial value. The value is therefore **UNSET**

```
!name = STRING()      $* To create a LOCAL, UNSET STRING variable

!!answer = REAL()     $* To create a GLOBAL, UNSET REAL variable

!array = BOOLEAN()    $* To create a LOCAL, UNSET ARRAY variable
```

2.3.2 Naming Conventions

It is common practice to follow a naming convention when defining objects and variables. Using upper case for the name of the object type and mixing upper and lower case for the variable is a good practice to follow.

For example, the type might be WORKERS while the name of the variable might be numberOf Workers.

Notice that to make the variable name more meaningful, full words are used with a mixture of upper and lower case letters to make it readable.

Variable names should not start with a number or contain any spaces or full stops (full stops are used in PML2 to indicate methods and members – explained later).

ⓘ *AVEVA uses a CD prefix on most of its global variables. Newer functionality does not use a prefix so all new PML must be checked for name clashes. Using your own prefix could help avoid this.*

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

2.3.3 Using the Members of an Object

When a variable is declared as an object-type, it is also given all the objects members and methods. For example, to a local variable as a factory object:

```
!newPlant = object FACTORY()
```

After being declared as above (using the OBJECT keyword and ending with a double bracket, the local variable !newPlant is now a FACTORY object and has the same members as the FACTORY object. These members are available to store information and can be assigned values in the following way:

```
!newPlant.name = |ProcessA|
!newPlant.workers = 451
!newPlant.output = 2000
```

ⓘ *Notice the use of a dot between the variable and its member. This works as long as the word after the dot is a valid member of the variable object-type.*

Once assigned a value, this value is available for use and can be recalled. For example:

```
!numberOfWorkers = !newPlant.workers.
```

This creates a new local real variable and assigns it the value 451.

2.3.4 Special Objects Used in E3D

In a standard E3D DESIGN session, there are number of specialised objects which are loaded and used by standard product. These objects should not be deleted or overwritten, but are available for use. The particularly useful objects are:

!!CE	- a global DBREF object which tracks and represents the current element
!!ERROR	- a global ERROR object which holds information about the last error
!!PML	- used to obtain file path strings through the .getPathName() method
!!ALERT	- used to provide popup feedback to users
!!AIDNUMBERS	- used to manage aid graphics
!!APPDESMAIN	- the form which represents the main DESIGN interface

2.4 PML Functions and Methods

Functions and methods provide the actions of PML. When called, a function or method will run through the lines of PML it contains in order – just like a PML 1 style macro. Functions and methods may be passed arguments and even return values. A function which does not return a value is typically referred to as a PML Procedure.

A function and a method are written in the same style, the only difference is where the definition is stored and how it is called. A function is a global method (stored in its own file) and can be called directly on the command line (e.g. call !!exampleFunction()) while a method is local to the object it is defined within (e.g. !exampleObject.exampleMethod()).

Arguments become local variables within the function/method and the object-types need to be declared within the definition. The returned object-type is also defined. For example:

```
define function !!area( !length is REAL, !width is REAL) is REAL
  !area = !length * !width
  return !area
endfunction
```

In this example, the function !!area is expecting two real arguments. The two arguments are expressed as local variables which are multiplied together to calculate the local variable !area. Using the **return** keyword,

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

the variable !area is then returned. If the function was called in following way, the variable !area will have a value of 2400:

```
!area = !!area(12, 200)
```

ⓘ *It is now common practice to write all macros as a functions or procedures*

2.4.1 Arguments of Type ANY

When defining an argument within a method or function, you may declare it as an ANY object. This means that the argument can be of any object-type, allowing any variable to be supplied to the function. As no argument check is carried out, the function needs to be robust enough to cope with any argument. For this reason, ANY should be used with special consideration:

```
define function !!argumentType(!argument is ANY) is STRING
  !type = !argument.objecttype()
  return !type
endfunction
```

2.5 PML Forms

For users of E3D, the concept of a form should already be familiar. In terms of PML, a form is a global variable (for example, !!exampleForm). A form is capable of owning members and methods (like any other object) but there are a set of predefined members which can be used to put gadgets on the form (buttons, lists, options etc). These gadgets are objects in their own right and can be given callbacks, which can activate a standard command or call a function or run a method defined within the form.

```
setup form !!nameCE
  !this.formTitle = |Name CE|
  button .button |Print Name Of CE| call |!this.print()|
exit
define method .print()
  !name = !!ce.flnn
  $p Name of Current Element = $!name
endmethod
```

The above example is the definition of a form called **nameCE**. Saved within one file (.pmlfrm), it defines two form members (a predefined member for the title and a new button gadget) and a form method.

!this is a special local variable and using it replaces the need to reference the owning object directly. For example, to call the method .print() from within the form, the call is !this.print() and to call the method from anywhere else, it's !!nameCE.print().

2.6 E3DUI Environment Variable

All PML1 Macros are typically stored a directory structure pointed at by the E3D environment variable E3DUI. The E3DUI environment variable is set in the .bat file that is used to load E3D (typically within evars.bat). To set the environment variable, the following line is needed in the .bat file:

```
set E3DUI=C:\AVEVA\Plant\E3D12.1.1\E3Dui
```

The purpose of an environment variable is to reduce the length of the command used to call a macro. This means that $M/%E3DUI%\DES\PIPE\MPIPE can be typed instead of the full file path.

In standard product, this process is shortened further as all PML1 macros and forms are called using synonyms. For example, the macros associated with piping are called using the synonym CALLP:

```
$S CALLP=$M/%E3DUI%/DES/PIPE/$s1
CALLP MPIPE
```

ⓘ *If all synonyms are killed then E3D will cease to function as normal*

Copia para EL AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

2.7 PMLLIB Environment Variable

All PML 2 objects and functions are in a directory structure pointed at by the E3D environment variable PMLLIB. As with the E3DUI variable, this is set in the .bat file which runs E3D. To set the environment variable, the following line is needed in the .bat file:

```
set PMLLIB=C:\AVEVA\Plant\E3D12.1.1\pmllib
```

The PMLLIB environment variable differs because it can be searched dynamically. This means that the individual files do not need to be referenced directly and can be called by name. This is possible because E3D compiles a pml.index file which sits in the PMLLIB folder and provides the path to all suitable files within it. For example, to load and show a form the command is **`show !!exampleForm`**, there is no need to reference the file path at all.

If a new file is created or the PMLLIB variable is changed then there is a need to update the pml.index file. This can be done by typing **`PML REHASH`** onto the command window. If there are multiple paths that need updating, type **`PML REHASH ALL`**

If a PML object has already been loaded into E3D, but the file definition has changed then the object needs to be killed and reloaded before the changes can be seen. This can be done by typing either **`pml reload form !!exampleForm`** or **`pml reload object EXAMPLEOBJECT`**.

Although not necessary, it is good practice to organise the files below the PMLLIB folder. A standard E3D installation organises the files based on application and then on forms, functions, objects. This is normally a good starting point for organising customisation.

2.8 Modifications to the E3DUI and PMLLIB

As PML is developed, it is normal practice to store it in a parallel file structure outside the standard installation directory. This parallel file structure can then be reference by the .bat files. There are a couple of reasons why this is a good idea:

- It keeps the customisation separate from the standard install, so as subsequent versions are installed there will not be a need to move the customised files.
- If any standard files are modified then the originals are still available if required.
- If the customisation fails, the standard installation is available to go back to.
- Many local E3D installations can reference the same customisation from a network address.

ⓘ *Any changes to AVEVA standard product may cause E3D to function inappropriately*

It is possible to get E3D to look in different places for PML and this is done by setting the environment variables to multiple paths. This allows the standard install to be kept separate from user and company customisation. This is done by updating the variable to include another path, for example:

```
set E3DUI=C:\temp\E3Dui %E3Dui%
set PMLLIB=C:\temp\pmllib %pmllib%
```

This will put the additional file path in front of the standard (which would have already been defined in the .bat file). This change can also be checked in a E3D session by typing **`q evar E3DUI`** or **`q evar PMLLIB`** onto the command window.

copia para EB AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Exercise 1 – Updating the Environment Variables

- Extract the provided files into the folder a suitable folder, for example C:\temp\
- Right –click on the icon that opens E3D. Choose Open File Location



- This can also be found from the Properties… form

- Open the file in a suitable text editor
- Add the following lines to **E3D.bat** (somewhere after the line that calls **evars.bat**)
  set E3DUI=c:\temp\E3Dui %E3Dui%
- `set PMLLIB=c:\temp\pmllib %pmllib%`
- Save the .bat as a new file to the computers desktop. This is now the icon you will use to enter E3D.

- Load E3D by using the new .bat file.
- The choice of project and MDB will depend on what is available in the E3D setup. This should be discussed with your trainer.

- Confirm that the search paths have been correctly updated by typing `q evar E3DUI or q evar PMLLIB` into the command window.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

This page is intentionally left blank.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

3 Macros, Synonyms and Control Logic

A macro is a group of E3D commands written to a file (in sequence) that can be run together. This removes the need for user to have to enter every line of code separately and that repetitive processes can be run as a macro.

3.1 A Simple Macro

ⓘ *The following example is a macro which creates an EQUI element which owns a BOX and CYLI. To test the example drag and drop the provided file into the command window.*

ⓘ *When running this macro in, ensure the Current Element (CE) is a ZONE or below.*

```
NEW EQUIP /ABCD
NEW BOX
XLEN 300 YLEN 400 ZLEN 600
NEW CYL DIA 400 HEI 600
CONN P1 TO P2 OF PREV
```

To run the macro into E3D, either drag and drop the file into the command window or type $M/ followed by the full file path of the saved macro.

e.g. `$M/%E3DUI%\examples\simpleMac.mac`

3.2 Finding Examples of Command Syntax

While developing a macro there are four main techniques of deriving the command syntax needed:

- DB listing utility — create the required elements in E3D using the standard appware and use the DB Listing utility to output the information (Utilities>DB Listing…)
- $Q syntax — if part of a command is known, the syntax which follows it can be found by typing `$Q` after the command e.g. `NEW $Q`
- Using standard product — standard E3D is supplied with numerous PML files which can be searched for keywords and used for inspiration
- Reference Manuals — the reference manuals supplied with the product focus on command syntax and supply syntax graphs for each case.
- As macros develop in complexity and use, it is likely that a mixture of the above will be used.

3.3 Communicating with AVEVA Products in PML

All commands need to be supplied to the command processor as STRINGs. This is important when working with variables as they may have another object-type. If a $ symbol is put in front of a variable, E3D will expand the contents of the variable as a string and then read the line. For example:

```
!componentType = |BOX|
!xLength = 5600
NEW $!componentType XLEN $!xLength
```

This is the equivalent of writing `NEW BOX XLEN 5600`

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

3.4 Parameterised Macros

Macros can be parameterised. This means instead of hard coding the values in the macro, arguments can be passed to it and used instead. This allows the values to be varied, making the macro flexible. As an example, simpleMac.mac could be parameterised in the following way:

```
NEW EQUIP /$1
NEW BOX
XLEN $2 YLEN $3 ZLEN $4
NEW CYL DIA $3 HEI $4
CONN P1 TO P2 OF PREV
```

To run this Macro, parameters need to be passed to it. The four parameters are defined by including the values of the parameters after the macro call:

e.g. **$M/%E3DUI%\examples\parameterMac.mac ABCDE 300 400 600**

If this macro is dragged and dropped into the command window, the parameters would be undefined and the macro would fail. To avoid this, default parameter values can be set at the top of the macro using the $d= syntax. For example:

```
$d1=ABCDEF
$d2=300
$d3=400
$d4=600
```

ⓘ *If defined, the default values will only be used if no parameters are passed to the macro.*

Macros may have up to nine parameters separated by space. In the below example, ABC, DEF & GHK are seen as separate strings and therefore different parameters:

e.g. **$M/%E3DUI%\examples\nineParameter.mac ABC DEF GHK 55 66 77 88 99 00**

If a text string is required as a single parameter, it can be entered by placing a $< before and a $> after the string. $< $> act as delimiters and anything in between is interpreted as a single parameter.

e.g. **$M/%E3DUI%\examples\sevenParameter.mac $<ABC DEF GHK$> 55 66 77 88 99 00**

3.5 Synonyms

Synonyms are abbreviations of longer commands. They are created by assigning a command to a synonym variable. To call the command held by the synonym, just type the name of the synonym.

e.g. **$SNewBox=NEW BOX XLEN 100 YLEN 200 ZLEN 300** called by typing **NewBox**

It is also possible to parameterise a synonym:

e.g. **$SNewBox=NEW BOX XLEN $S1 YLEN $S2 ZLEN $S3** called by typing **NewBox 100 200 300**

A synonym can call itself and if it uses the $/ syntax (return character) it can be used to loop through elements. For example, a new XLEN value can be applied to all elements for level in the hierarchy.

e.g. **$SXChange=XLEN 1000 $/ NEXT $/ XChange**

Synonyms can be turned off and on with the **$S-** and **$S+** syntax. To kill a synonym, type **$SXXX=** and to kill all synonyms `$sk`. Standard E3D relies on synonyms to function.

ⓘ *If a required synonym is killed, the product will no longer function properly (requiring a restart)*

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

3.6 Defining Variables

There are number of different methods for defining variables in E3D:

3.6.1 Numbered Variables

Numbered variables are set by typing a number then value after the command VAR. Examples of this are below:

```
var 1 name
var 2 |hello|
var 3 (99)
var 4 (99 * 3 / 6 + 0.5)
var 117 pos in site
var 118 (name of owner of owner)
var 119 'hello ' +  'world ' +  'how are you'
```

The value of variable 1 can be obtained by typing `q var 1` into the command window. The available variable numbers only go to 119 (there is no 120) and they are module dependant.

ⓘ *This technique is no longer commonly used and has been included for completeness.*

3.6.2 PML 1 Style Variables

The following are some examples of setting variables using the PML 1 style syntax:

```
VAR !NAME NAME                  $* Takes the current element's (CE) name attribute
VAR !POS POS IN WORLD           $* Takes CE position attribute relative to the world
VAR !x |NAME|                   $* Sets the variable to the text string |NAME|
VAR !temp (23 * 1.8 + 32)       $* Calculate a value using the expression
VAR !list COLL ALL ELBO FOR CE $* Makes a string array of database references
```

This style of defining variables is still valid and is useful for command syntax that returns a value. It has been included within this guide primarily for information for when old code is upgraded.

3.6.3 PML 2 Style Variables

The following are some examples of setting variables using the PML 2 style syntax:

```
!name = !!ce.name               $* Takes the current element's (CE) name attribute
!pos = !!ce.pos.wrt(/*)         $* Takes CE position attribute relative to the world
!x = |NAME|                     $* Sets the variable to the text string |NAME|
!temp = 23 * 1.8 + 32           $* Calculate a value using the expression
```

These examples show that there are equivalents to the PML 1 style, but because PML 2 is object-based this can be incorporated into the declaration of the variable.

The equivalent PML2 collection will be explained later in the training guide.

3.7 Expressions

Expressions are calculations using PML variables. This can be done in a PML 1 or PML 2 style:

```
VAR !Z ( |$!X| + |$!Y| )        PML 1
!Z = !X + !Y                    PML 2
```

ⓘ *In the PML1 example, !Z is set as a STRING variable. In the PML2 example, !Z is returned as a REAL, if !X and !Y are REAL*

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Copia para EE AA

3.7.1 Expression Operators

There are a number of expression operators which are available for use within PML. The numeric operators and functions are the same for both PML 1 and PML 2 styles. For comparison and logic operators there are new PML 2 methods available:

Numeric operators:	+ - / *
Numeric functions:	SIN COS TAN SQR POW NEGATE ASIN ACOS ATAN LOG ALOG ABS INT NINT
PML 1 style Comparison operators:	LT GT EQ NE LE GE
PML 1 style Logic operators:	NOT AND OR
PML 2 style Comparison methods:	.lt() .gt() .eq() .neq() .leq() .geq()
PML 2 style Logic methods:	.not() .and() .or()

The PML 2 comparison methods are available on any object-type variable, providing it is compared to a variable of the same type. The PML 2 logic methods are available on the BOOLEAN object

ⓘ *For further examples, refer to E3D Software Customisation Reference Manual:*

Some examples of expressions in use:

!s = 30 * sin(45)

!t = pow(20,2) (raise 20 to the power 2 (=400))

!f = (match(name of owner,|LPX|)gt 0)

3.7.2 Operator Precedence

When E3D reads an expression, there is a precedence applied to it. This should be considered when writing expressions. The order is as follows:

()	e.g.	60 * 2 / 3 + 5	= 45
*** /**		60 * (2 / (3 + 5))	= 15
+ -			
EQ NE GT LT GE LE			
NOT			
AND			
OR			

3.7.3 PML 2 Expressions

PML 2 expressions may be of any complexity and may derive the required values from PML Functions and Methods and include Form gadget values, object members and methods. For example:

!newValue = !!myFunc(!OldVal) * !!form.gadget.val / !myArray.method()

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

3.8 Arrays

An ARRAY variable can contain many values, each of which is an ARRAY ELEMENT.

An array is created when an array element is defined or it can be initialised as an empty array (i.e. !x = ARRAY()). If an ARRAY ELEMENT is itself an ARRAY, this will create a Multi-dimensional ARRAY. For example, type out the following onto the command window to define an array:

!x[1] = |ABCD|
!x[2] = |DEFG|
!y[1] = |1234|
!y[2] = |5678|
!z[1] = !x
!z[2] = !y

To query the information about !z, type **q var !z**. This will return the following information:

```
<ARRAY>
   [1]  <ARRAY> 2 Elements
   [2]  <ARRAY> 2 Elements
```

To find out more information about the elements within the Multi-dimensional array, type **q var !z[1]** or **q var !z[2][1]**

3.9 Concatenation Operator

Values are automatically converted to STRING and concatenated when the '&' operator is used. Type the following onto the command window and compare this against the results of typing **!d = !a + !b**

!a = 64
!b = 32
!m = |mm|
!c = !a & !b & !m
q var !c

3.10 DO Loop

A DO loop is a way of repeating PML, allowing pieces of code to be run more than once. This is useful as it allows code to be reused and reduces the overall number of lines. As an example, try the provided file %E3DUI%\examples\doLoop.mac:

```
DO !loopCounter TO 10
  !value = !loopCounter * 2
  q var !loopCounter !value
ENDDO
```

In the above example, as the loop runs, values of !loopCounter and !value with be printed to the command line for the full range of the defined loop. The step of the loop can be altered by adding FROM and BY to the loop definition. For example, try the provided file **%E3DUI%\examples\doLoopMax.mac**:

```
DO !loopCounter FROM 5 TO 10 BY 2
  !value = !loopCounter * 2
  q var !loopCounter !value
ENDDO
```

3.10.1 DO Loops with BREAK

If you need a loop to run until a certain condition is reached, a BREAK command will exit the current loop. This can be used in conjunction with an infinite loop or with a loop with a range. As an example, try the provided file %E3DUI%\examples\doLoopBreak.mac:

```
DO !n
  !value = POW(!n, 2)
  q var !value
  BREAK IF (!value GT 1000)
```

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

```
ENDDO
```

The loop in the example will run until the BREAK condition is met. If the condition is never reached, then the code will run indefinitely!

> ⓘ *It is good practice to give a loop a range, even if very large (i.e. 1 to 100000). This ensures that the loop has an end and won't require E3D to be crashed to exit it.*

The BREAK command can also be called from within a normal IF statement. This is typically done if multiple break conditions need to be considered or if additional code needs running. For example:

```
IF (!value GT 1000) THEN
  !!alert.message(|1000 reached|)
  BREAK
ENDIF
```

3.10.2 DO Loops with SKIP

It is possible to skip part of the DO loop using the SKIP command. This could be useful if parts of a number sequence needs to be missed. As an example, try the provided file %E3DUI%\examples\doSkip.mac:

```
DO !n FROM 1 TO 25
  SKIP IF (!n LE 5) OR (!n GT 15)
  q var !
ENDDO
```

The SKIP command can also be called within a normal IF statement (as with the BREAK command)

3.10.3 DO INDEX and DO VALUES

DO INDEX and DO VALUES are ways of looping through arrays. This is an effective method for controlling the values used for the loops. Typically values are collected into an ARRAY variable then looped through using the following:

DO !X VALUES !ARRAY $* !X takes each ARRAY element as its value
DO !X INDEX !ARRAY $* !X takes an incremental number from 1 to !ARRAY size

Try the provided file %E3DUI\examples\doArray.mac.

```
VAR !zones COLL ALL ZONES FOR SITE
VAR !names EVAL NAME FOR ALL FROM !zones
q var !names
DO !x VALUES !names
  q var !x
ENDDO
DO !x INDEX !names
  q var !names[!x]
ENDDO
```

3.11 IF Statements

An IF statement is a construct for the conditional execution of commands. The commands within the statement will only be run if the conditions are met. In the following example, the code within the IF construct is only run if the expression is TRUE (i.e. !number is real and less than 0):

```
IF ( !number LT 0 ) THEN
  !negative = TRUE
ENDIF
```

The expression can be written in any form, providing the answer is BOOLEAN. For example, the following is the same as above but written in a PML 2 style:

```
IF ( !number.lt(0) ) THEN
```

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

If the variable itself is already BOOLEAN, a comparison does not need to be made. For example:

```
!booleanVariable = TRUE
IF ( !booleanVariable ) THEN
```

Or `IF ( !!exampleFunction() ) THEN` - where !!exampleFunction() returns a BOOLEAN result

3.11.1 IF, ELSEIF and ELSE Statements

An IF statement is extended by adding additional conditions to it and this is done by using ELSEIF or ELSE statements. When an IF statement is encountered, E3D will evaluate its first condition. If the condition is FALSE, E3D will look to the next ELSEIF condition.

Once a condition is found to be TRUE, that part of the code will be run and then the IF statement is complete. If an ELSE condition is added, this portion of code will only be run if no other conditions are met. This is a way of ensuring some code runs as part of the construct. As an example, try the provided file %E3DUI\examples\numCheck.mac. Run the macro with one real parameter for the comparison.

```
IF ($1 EQ 0) THEN
  $p Your value is zero
ELSEIF ($1 LT 0) THEN
  $p Your value is less than zero
ELSE
  $p Your value is greater than zero
ENDIF
```

ⓘ *The ELSEIF and ELSE commands are optional, but there can only be one ELSE command in an IF construct.*

3.12 Branching

PML provides a way of jumping from one part of a macro to another using the GOLABEL syntax.

```
LABEL /FRED
...
Some PML code
...
GOLABEL /FRED
```

The next line to be executed after GOLABEL /FRED will be the line following LABEL /FRED, which could be before or after the GOLABEL command. The name of the label can be up to 16 characters (excluding the leading slash)

ⓘ *The use of this method should be limited as it can make code hard to read and therefore to debug.*

3.12.1 Conditional Branching

It is possible to add a condition to the GOLABEL syntax such that the jump will only be made if the condition is TRUE. As an example, type out the following and save it as %E3DUI%\examples\conditional.mac:

```
DO !A
$P Processing $!A
  DO !B TO 10
    !C = !A * !B
    GOLABEL /finished if (!C GT 100)
    $P Product $!C
  ENDDO
ENDDO
LABEL /finished
$P Finished with processing = $!A Product = $!C
```

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

If the expression **!C GT 100** is **TRUE** there will be a jump to label /**finished** and PML execution will continue with the **$P** command. If the expression is **FALSE**, PML execution will continue with the command: **$P Product $!C** and go back through the DO loop.

3.13 Error Handling

An error condition can occur when a command could not complete successfully. This is because of a mistake in the executed macro or function. An error normally has three effects:

- An Alert box appears which the user must acknowledge.
- An error message is outputted to the command line together with a trace back to the error source.
- Any current running PML macros and functions are abandoned.

3.13.1 Error Codes

When an error occurs, an error code is returned to the user along with a message. The following example error is caused when trying to create an EQUI element in the wrong part of the hierarchy.

(41,8) ERROR – Cannot create an EQUI at this level.

Where 41 is the program section which identified the error and 8 is the error code itself.

3.13.2 Error Handling Using the HANDLE Syntax

If the input line which caused the error was part of a PML macro or function, the error may optionally be **HANDLED**. This allows the designer of the macro to limit the errors the user will experience.

As an example, try the provided file %**E3DUI%\examples\errorTest.mac**. First run the macro at a **SITE** element, then at a **ZONE** element and then again at the same ZONE. Compare the return printed lines in the command window.

```
NEW EQUI /ABCD
HANDLE (41, 8)
  $p Need to be at a ZONE or below
ELSEHANDLE (41, 12)
  $p That name has already been used. Names must be unique
ELSEHANDLE ANY
  $p Another error has occurred
ELSEHANDLE NONE
  $p Everything OK. EQUI created
ENDHANDLE
```

ⓘ *Notice how the HANDLE syntax can differentiate between specific error codes and how it is possible to capture all errors or only run if no error occurs.*

3.14 Alert Objects

Alert objects allow user-controlled pop-up forms to be used, providing feedback to the user. The choice on which form type is used and whether one is needed at all is down the PML designer.

3.14.1 Alert Objects with no Return Value

There are three types of alert with no return value:

```
!!alert.error( |You cannot do this!| )
!!alert.message( |Saving your data now| )
!!alert.warning( |Incorrect value!| )
```

Copia para FEL AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.



By default, all alert forms appear with the relevant button as near to the cursor as possible. To position an alert specifically, X and Y values can be specified as a proportion of the screen size.

!!alert.error(|You cannot do this!| , 0.25, 0.1)

3.14.2 Alert Objects that return value

There are three types of alert which return a value. This value can be subsequently used to as an indication of the decision the user has made (i.e. as part of an IF statement).



3.14.2.1 Confirm Alerts

A Confirm alert returns a string of 'YES' or 'NO'

!answer = !!alert.confirm(|Are you sure!|)
q var !answer



3.14.2.2 Question Alerts

A Question alert returns a string of 'YES' or 'NO' or 'CANCEL'

!answer = !!alert.question(|OK to delete Site?|)
q var !answer

3.14.2.3 Input Alerts

An Input alert returns the value entered by the user. If the user does not enter a value, then the default value (set at definition) is returned.



!answer = !!alert.input(|Enter Width of Floor|, |100|)
q var !answer

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

3.15 Undo and Redo

If you build an application that creates or deletes items from the E3D Database it is good practice to handle the ability to undo the modifications. Undo is a useful feature that users expect to be able to use.

3.15.1 Marking the Database

The undo button goes back to the last database mark or savework so your application should mark the database before modification.
MarkDB |Comment|

3.15.2 Undo and Redo Database Commands

UndoDB - Undo the database back to the last database mark

RedoDB - Redo the last Undo

These actions can be performed by including the above lines within your code. There is also an UNDOABLE object with built-in methods which can extend this functionality further.

Refer to the Software Customisation Reference Manual for information about the UNDOABLE object

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Exercise 2 - The Centre of a Handwheel

- Write a macro that will produce the centre of a handwheel (shown to the right)
- It shall be constructed from 4 primitives: 2 boxes, a cylinder and a dish and should create the following hierarchy:



- The macro shall create the primitive with the following fixed sizes:

First BOX	X Length = 100mm; Y Length = 100mm; Z Length = 100mm
CYLI	Diameter = 80mm; Height = 5mm
Second BOX	X Length = 50mm; Y Length = 50mm; Z Length = 15mm
DISH	Diameter = 50mm; Height = 15mm

- The primitives should be unnamed, but sit below an EQUI called HandWheel

- Let the first BOX be created without specifying a position or orientation.
- Use the CONN syntax to position the next primitive to the first BOX. For example:

  CONN P1 TO P1 OF PREV

- This will connect PPoint 1 (P1) of the current primitive to P1 of the primitive created previously.
- To help with connecting the primitives, refer to the adjacent diagram showing an exploded version of the equipment, with the PPoints identified.



- You may wish to add some other syntax to the macro:

  To add the HandWheel to the drawlist

  ADD /HandWheel

  To set the limits of the view to the HandWheel

  AUTO /HandWheel

  To completely clear the drawlist

  REM ALL



copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

- Save the file as c:\temp\E3Dui\ex2.mac and run it on the command line by typing **$m/%E3DUI%\ex2.mac** or by dragging and dropping the file into the command window.

An example of the completed macro can be found in Appendix B

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Exercise 3 - The Full Handwheel

- Extend the previous macro to build the remaining parts of the HandWheel (shown below).



- Information about primitives can be found in Appendix A

- It is possible to build the remaining primitives individually, but as there is a rotational centre, a copy-rotate could be used.
- To copy a BOX element (where /XXXX is another BOX. Note: the word PREV is also valid)

  NEW BOX COPY /XXXX
- To rotate the current element (where AAA is the angle and BBB the direction)

  ROTATE BY AAA ABOUT BBB

- To help with organization of the primitives of the EQUI, you may wish to add additional SUBE elements. This could help with the copy-rotate methods
- An example hierarchy is shown to the right, but this will depend on your macro



- Turn the macro into a parameterized macro. Pass it two parameters (1) the name of the EQUI and (2) the outside diameter of the HandWheel
- Think about how the sizes of the primitives will relate to the outside diameter. You may need to do some calculations before the value can be used
- As parameters are used, set the default values

- If any errors occur in your macro, consider error handling or changing the macro so that the error cannot happen
- Save the file as **c:\temp\E3Dui\ex3.mac** and run it on the command line by typing **$m/%E3DUI%\ex3.mac** or by dragging and dropping the file into the command window

An example of the completed macro can be found in Appendix B

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

This page is intentionally left blank.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

4 PML Functions

Functions are lines of PML grouped together in a file designed to do something. When called, the lines of the function are run and the intended action completed.

4.1 Functions Instead of Macros

Functions are designed to replace macros and can contain the same syntax that would have been placed inside a PML 1 style macro. It is intended that macros should no longer be used and any future PML should be written as functions. The following are the reasons why:

- Functions are preloaded through the PMLLIB searchpath.
- Functions can accept arguments of specific object-types.
- Functions can return values of specific object-types.

4.2 Creating a PML Function

A function is saved as a .pmlfnc file under the PMLLIB search path. The filename should be the same name as the function, for example, if the function is called !!exampleFunction then the file should be called exampleFunction.pmlfnc.

A function will typically return a value, although arguments are optional. The object-type of the returned object needs to be defined, and is done so at the end of the define function line.

The following is an example of a simple function designed to return the full name of the current element. It is an example of a RETURN function with NO ARGUMENTS.

```
define function !!nameCE() is STRING
  !ce = !!ce.flnn
  return !ce
endfunction
```

After it has been saved as a .pmlfnc file and a **PML REHASH ALL** on the command line, the function can be called by typing:

!name = !!nameCE()

After running this line, variable !name will be a string holding the full name of the current element.

If a function is defined with arguments, then these arguments will become local variables available within the function. These can then be used in expression, to control the function etc. The following is an example of a RETURN function with an ARGUMENT:

```
define function !!circleArea(!radius is REAL) is REAL
  !circleArea = !radius.power(2) * 3.142
  return !circleArea
endfunction
```

After definition, the function can be used as part of an expression as it returns a real object. This means that common expressions/calculations can be written as a function and used as required.

!height = 64
!cylinderVolume = !!area(2.3) * !height
q var !cylinderVolume

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

4.3 PML Procedures

A function that does not return a value is typically described as a PML procedure. This means that as there is no return and the return object-type does not need to be defined. The following example of a NON-RETURN function with NO ARGUMENTS:

```
define function !!lockCE()
  !!ce.lock = !!ce.lock.not()
endfunction
```

The following is an example of a NON-RETURN function with ARGUMENTS:

```
define function !!setBoxPrimitiveSize(!x is REAL, !y is REAL, !z is REAL)
  if !!ce.type.eq(|BOX|) then
    !!ce.xlen = !x
    !!ce.ylen = !y
    !!ce.zlen = !z
  endif
endfunction
```

Notice how both the examples use the !!CE object (a global object which represents the current element). PML procedures are typically used to interact with defined global variables

4.4 Recursive PML2 Functions

PML functions may be called recursively. This can be used to produce iterative solvers.

```
define function !!fibonacci(!last is REAL, !previous is REAL, !array is ARRAY)
  !next = !last + !previous
  !localArray = !array
  !localArray.append(!next)
  if !localArray.size().lt(50) then
    -- Here the function calls itself
    !!fibonacci(!next, !last, !localArray)
  else
    q var !localArray
  endif
endfunction
```

This function shows how an array of Fibonacci numbers (an increasing series of numbers, where the next number is the sum of the previous two) can be compiled.

4.5 Making Use of Methods on PML Objects

After a variable has been defined as a specific object-type, the methods of the object will be available on the variable. Making use of these methods can help manipulate the information within a function to ensure the correct outcome.

From the example on the previous page, !radius is defined as a REAL object. Therefore it has access to all the methods of a real object. The example makes use of the .power() method, and method which takes a real argument and raises the !variable to the power of the argument. The method returns the answer as a real object.

When working with built-in objects, refer to E3D Software Customisation Reference Manual for the available methods and information about them

If you are working with different object-types, it is possible to switch been types. For example, there may be a need for a STRING to be seen as a REAL for use in an expression. The following would give an error:

```
!value = |56|
!result = !value * 2
```

Copia para FEI AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

As a STRING object cannot be multiplied by a REAL, an error will be returned. To avoid this error, the .real() method can be used to return a STRING from a REAL. Note, the original variable remains a **STRING**, but it is seen as a **REAL** for the expression:

```
!value = |56|
!result = !value.real() * 2
```

ⓘ *There are equivalent methods available for the other standard objects, refer to E3D Software Customisation Reference Manual.*

4.5.1 Method Information

For each object-type in the Software Customisation Reference Manual, there is a table which lists the available methods and supporting information. For each method there will be:

NAME — The name of the method or member. For example, a REAL object has a method named Cosine.

If there are any arguments, they are indicated in the brackets () after the name. For example, the REAL object has a method named BETWEEN which takes two REAL arguments

RESULT — The type of value returned by the method. For example, the result of the method Cosine is a REAL value. Some methods do not return a value: these are shown as NO RESULT.

PURPOSE — This column tells you what the member or method does along with other information about the method or members.

ⓘ *Note that for the system-defined E3D object types, the members can be listed by using the **.attributes()** method and the methods can be listed using the **.methods()** method*

The following are some examples of ARRAY object methods to explain the different styles:

!numberOfNames = !nameStrings.size()

This method returns the number of elements currently in the array. This is an example of a RESULT method with NO-AFFECT on the original object.

!nameStrings.clear()

This method deletes the contents of the array, but not the array. This is an example of a NO RESULT method which does AFFECT the original object.

!newNameArray = !nameStrings.removeFrom(5,10)

This method result removes 10 elements (starting at element 5) from the array. These elements are then returned by the method. This is an example of a RESULT method which does AFFECT the original object. If you need to remove part of the array and do not need it, then it does not need assigning to a variable (e.g. type **!nameStrings.removeFrom(5,10)**)

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

4.5.2 Method Concatenation

One of the advantages of object-orientated programming is that methods can be built-up within a single line of code. This means that complex manipulations can be made in fewer lines of PML. This process will only work if the previous method returns a suitable object for the following method. For example:

```
!line = 'hello world how are you'
!newline = !line.upcase().split().sort()
q var !line !newline
```

```
<STRING> 'hello world how are you'
<ARRAY>
      [1] <STRING> 'ARE'
      [2] <STRING> 'HELLO'
      [3] <STRING> 'HOW'
      [4] <STRING> 'WORLD'
      [5] <STRING> 'YOU'
```

In this example, the first two methods are valid for STRING objects. The .split() method returns an ARRAY object so the following method has to be a valid method for an ARRAY object.

4.6 Using the !!CE Object

!!CE is a special GLOBAL PML variable that always points to the current E3D element. It is a DBREF object and its members are the attributes of the current element. Type **q var !!CE** onto the command line and compare it against the elements attributes (Query>Attributes...). You will notice the returned attribute information is the same of the members list of the !!CE object. This means that the !!CE object can be used to assign the values of attributes to !variables. For example:

```
!branchHeadBore = !!CE.hbore
```

This assigns the HBORE attribute (taken from the current BRAN element) to the variable !BranchHeadBore making it real.

ⓘ *It will be necessary to check that HBORE is a valid attribute of the current element before running this line. It may cause an error.*

If the !!CE object member is an object itself, that object could also have members so further information be obtained. For example, obtain the east coordinate of a head of a BRAN element, either of the two following can be used:

```
!headPosition = !!CE.hpos
!headEasting = !headPosition.east
```

Or: **!headEasting = !!CE.hpos.east**

If the !!CE object member is an object with built-in methods, then these methods can also be called:

!PosWRTValve = !!CE.hpos.wrt(ZONE) returns a POSITION object w.r.t the owning ZONE

This process can also be reversed and values can be applied to the attributes of the !!CE. This means that it is possible to record the current value of an attribute, modify and reassign back to the CE. For example, type out the following onto the command line:

```
q POS
!position = !!CE.pos
!position.up = !position.up + 2000mm
!!CE.pos = !position
q POS
```

These lines will have moved the CE up by 2000mm. This same logic can be applied to other attributes, providing the object-type is considered.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Exercise 4 – Convert the HoseReel Macro into a PML Procedure

- You have been provided with an old PML 1 style macro that builds a Hose Reel as a piece of equipment. The macro is old and has not been maintained. There are at least 5 problems with it.
- Debug the existing macro and convert it into a new PML procedure with NO ARGUMENTS





- Save the new file as ex4.pmlfnc under the PMLLIB searchpath.
- You may wish to create a new folder to keep the exercises separate from the examples
- Update the PMLLIB search path by typing **PML REHASH ALL**
- Test the function in the command window by typing **!!ex4()**

- Update the procedure and give it TWO ARGUMENTS so it becomes a function**,** The two arguments should be:
  - The name of the EQUI
  - The diameter of the Hose Reel.
- Check the updated procedure to make sure the Hose Reel is created as expected. Consider error handling while you check the procedure.
- Save the modified procedure as another PML function file with a suitable name.

An example of the completed procedure can be found in Appendix B

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

This page is intentionally left blank.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

5 Collections

A very powerful feature of the E3D database is the ability to collect and evaluate data according to rules. There are two available methods for collection that are valid in PML. The first is a command syntax PML 1 style and the other is with a PML COLLECTION object. Both are still valid, but when developing new PML, a COLLECTION object should be used in preference.

5.1 COLLECT Command Syntax (PML 1 Style)

The COLLECT syntax is based around three specific pieces of information:

- What element type is required?
- If specific elements are required, what is different about them?
- Which part of the hierarchy to look it?

If you wish to collect all the EQUI elements for the current ZONE, type the following:

```
var !equipment collect all EQUI for ZONE
q var !equipment
```

If you wish to collect all the piping components owned by a specific BRAN, type the following:

```
var !pipeComponents collect ALL with owner eq /200-B-4-B1 for SITE
q var !pipeComponents
```

If you wish to collect all the BOX primitives below the current element, type the following:

```
var !boxes collect all BOX for ce
q var !boxes
```

ⓘ *You do not need to specify level of the hierarchy to search within i.e. the FOR.*

For more examples, refer to Chapter 11 of the Database Management Reference Manual and 2.3.10 Design Reference manual general commands.

5.2 EVALUATE command syntax (PML 1 style)

After elements have been collected through the COLLECT syntax, they are stored as an ARRAY. This array (like any array) can be processed using the EVALUATE syntax to provide further information

To get the names of all the elements held in !equipment, type the following:

```
var !equipmentNames evaluate NAME for ALL from !equipment
q var !equipmentNames
```

To get the full names of the elbows held within !pipeComponents, type the following:

```
var !elbows evaluate FLNN for all ELBO from !pipeComponents
q var !elbows
```

To get the names (without the leading slash) of the pumps in !equipment, type the following:

```
var !pumps evaluate NAMN for ALL with MATCHWILD(NAMN, |P*|) from !equipment
q var !pumps
```

To get the volume of all the boxes in !boxes, type the following:

```
var !volume eval (xlen * ylen * zlen) for ALL from !boxes
q var !volume
```

ⓘ *Notice how the expressions are command syntax and must return a BOOLEAN. Refer to the Software Customisation Reference Manual for more examples. COLLECTION object (PML 2 style)*

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

5.3 COLLECTION Object (PML 2 Style)

A COLLECTION object is another example of a PML object that has directly replaced some command syntax. It is recommended that all new PML uses COLLECTION objects as required. One advantage of using a COLLECTION object is that an ARRAY OF DBREF objects is returned.

A COLLECTION object is assigned to a variable in the same way as other objects:

```
!collection = object COLLECTION()
q var !collection
q var !collection.methods()
```

Once assigned, the methods on the object are used to set up the parameters of the collection. To set the element type for the collection to only EQUI elements, type the following:

```
!collection.type(|EQUI|)
```

If more than one element type is required, the following could be typed:

```
!elementTypes = |EQUI BRAN SCTN|
!collection.type(!elementTypes.split())
```

The scope of the collection must be a DBREF object and should be passed as an argument to the .scope() method. For example, type the following to set the scope as the current element:

```
!collection.scope(!!ce)
```

To filter the results, the .filter() method must be passed an EXPRESSION object. This means that the process is in two steps:

```
!expression = object EXPRESSION(|name of owner eq '/200-B-4-B1'|)
!collection.filter(!expression)
```

Once the collection object has been set up, the .results() is used to return the collected elements as an ARRAY of DBREF objects.

```
!results = !collection.results()
q var !results
```

5.4 Evaluating the Results from a COLLECTION Object

After an ARRAY of DBREF objects has been generated from a COLLECTION object, the entire array can be evaluated using the .evaluate() method on an array object. The argument for the .evaluate() method must be a BLOCK object defined with the expression that needs evaluating.

To get an ARRAY of STRINGs holding the full names of the collected elements, type the following:

```
!block = object BLOCK(|!results[!evalIndex].flnn|)
!resultNames = !results.evaluate(!block)
q var !resultNames
```

Notice how the BLOCK object uses the local variable !evalIndex. This variable effectively allows the .evaluate() method to loop through the ARRAY. To get the positions of the collected elements, type the following:

```
!resultPos = !results.evaluate(object BLOCK(|!results[!evalIndex].pos|))
q var !resultPos
```

Instead of defining the block as a separate variable, this second example shows that the object can be defined within the argument to another object.

ⓘ *Notice how the evaluated ARRAY contains the correct object types generated from the evaluation*

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Exercise 5 – Collection

- You have been asked to develop a PML function that will apply colour to an element type below the current element
- Write the function to accept two arguments:
  - The element type to be highlighted - STRING
  - The highlight colour to use - REAL
- The function will need to use PML to collect the information required. Try to use the COLLECTION object with preference. How would the PML differ if a PML1 style collection was used instead?
- For ease, use the **ENHANCE** syntax to highlight the elements. The following is the syntax graph taken from section 4.6 of the Design Reference Manual – General Commands:



ⓘ *The PML2 way of highlighting elements is discussed in TM1402*

- Save the new file as highlighttype.pmlfnc under the PMLLIB searchpath.
- Update the PMLLIB search path by typing **PML REHASH ALL**
- Test the function in the command window. For example, the following function call **!!highlighttype(|CTOR|, 2)** would highlight the hosewheel as shown.



- Update the function so that it returns an array of the collected elements.
- It is good practice to keep track of any element which your PML highlights. This means that it can be unenhanced as required (rather than just typing UNENHANCE ALL)
- Test the function by querying the size of the returned array on the command line
- For example:

```
q var !!highlighttype(|CTOR|, 2).size()
```

<REAL> 16

ⓘ *An example of the completed procedure can be found in Appendix B*

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

This page is intentionally left blank.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

6 Miscellaneous

6.1 Error Tracing

It is possible to list all the commands that E3D runs when a PML operation is carried out. This list can either be printed to the command window, or written to an external file.

Type **$r109 /c:\log.log** onto the command line

where c:\log.log is the output file name (it shall be created if it does not exist)

Now every PML action will be written to the external file. The file will list every line of code and action carried out. Lines read will be indicated by a line number in square brackets. Lines not read will be indicated by a line number between round brackets. Entry and exit points between methods, functions and objects are indicated as well as any errors.

Type **$r110** onto the command line to print the same lines to the command window

ⓘ *Printing to the screen should only be done when a small number of lines are expected. The command window may run out of lines to display the information*

Once you are finished with error tracing, type $r to the command line. If the file is not closed, information will continue to be added to it

ⓘ *This will need to be typed before the output file can be opened*

- To find out more information about the $r command, type $HR into the command window.

6.2 PML Publisher

It is now possible to encrypt any files you create before sharing them. Once encrypted, the files can still be used in any compatible AVEVA program, but they are not easily read through a normal text editor. Encrypted files may be used without additional licenses, but the encryption utility described below is separately distributed and licensed.

ⓘ *Once encrypted, PML cannot be decrypted. A reference copy of all PML should be kept safe*

ⓘ *PML Publisher is a separately licensed product and access to it will depend on you E3D license*

If the example form !!nameCE was encrypted (from page 17) the following file would be created:

```
--<004>-- Published PML 1.0 >--
return error 99 'Unable to decrypt file in this software version'g
$** abdfe19b3008494b6399edda08b66004
$** MR+zhtg-egE2Ig9IiHSVmdPo08ChKexa7wbfcyODTbfjTFWU02pK3v4sXq5i
$** TKW3dEFRJCd60uSzaLXdc5fvLeOKqXO71uFlZ1vEsIOOHvq8viAwiys4rGXg
$** XLgFFVG7mpsmnFtrQDN3o51aiAgicFS6u08C7r8IaxUTUQAOdXeBmlp4TLXc
$** 9KR5LtAIugLrC9a7NxbF+0Hn-c5tOhUAEBG
```

📖 *Reading an encrypted file is slower than reading a decrypted on. Making use of the Buffer argument can help. Refer to the PML Publisher User Guide for more information*

6.3 General Notes on PML

- PML functions and methods run through their definitions in line order (control logic can be used to alter this order)
- Functions should be written in preference to macros

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

- Variables are instances of object definitions
- The return command can be used to exit a function/method as well as return a value
- PML files are ASCII and can be created/edited in any basic text editor
- PML 1 files are saved under **E3DUI** folder and PML 2 under the **PMLLIB** folder
- PML 2 objects have specific file extensions (.pmlfnc, .pmlobj and .pmlfrm)
- If new created, PML can be found by typing **pml rehash all**
- Once loaded, objects can be reloaded by typing **pml object reload OBJECT**
- Once loaded, forms can be reloaded by typing **pml object reload FORM**
- When declaring a string, text delimiters must be used. Either '**single quotes'** or |**vertical bars**|
- File paths of files can be obtained by using **!!pml.getPathName(|form.pmlfrm|)**
- The $ is used as an escape character and can be used to expand a variable before it is read as command syntax. If a $ symbol is actually required, then two must be entered
- To provide comments in statements with PML, one of the following can be used:
- For a comment line, start the line with two hyphens **--**
- For a comment at the end of a line, start the comment with **$***
- For a block comment, start with a **$(** and end with a **$)**
- Variable names are not case sensitive
- String comparisons are case sensitive
- Variable names can be 16 characters long and should not start with a number or contain a dot
- It is good practice to name variables in lower case, using upper case to separate words e.g. !stringLength
- It is possible to abbreviate some command syntax. The required part is shown in upper case with the syntax graphs e.g. **WIDth** means that **WID** is acceptable when declaring width

6.4 Menu Additions

It is possible to define user defined toolbars to run your macros:

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Worked Example – Creating a toolbar within Design



- Open the Customisation Form. Do this by right clicking on an empty part of the toolbar and choose **Customize**…



- On the Active Customization File drop down, choose **User**.
- This will save an changes to a specific user file (file path given on the right hand side of the form)



- Right click on the Command Bars heading in LHS window, and click on **New CommandBar**



- When the new CommandBar is selected you will now see a preview of the command bar in the top left of the form

Copia para Ferraa

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.



- Right click anywhere in the middle window in the form
- On the context menu, choose **New>Button** to create a new button for the toolbar.



- With the new button selected (end of list), look to the right window on the form
- Select the line titled **Command** and click the small **"…"** button which appears on the right. This will let us set the command.



- A command can be a core command or a PML command or a command class.
- Choose **Macro**, and type **!!ex4().** Close the form by pressing OK.



- The associated command is now
- Select the line titled **Icon**. Click the **"…"** button that appears to the right.
- Browse for the file **equi.png** (a supplied file) and click **Open**



- The icon is now displayed in the right hand window.
- Set the tooltip to **'Build HoseWheel'**

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.



- Select the new button in the middle window.
- Drag and Drop the new button beneath the new CommandBar object to make the association.
- Select the CommandBar object and notice the button is displayed in the preview.



- **Apply** and **OK** the form and the new Command Bar will be displayed in the main application.
- Test the button

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

This page is intentionally left blank.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Appendix A – E3D Primitives

Box (BOX)



Specific geometric attributes:

Xlength	Length parallel to X axis
Ylength	Length parallel to Y axis
Zlength	Length parallel to Z axis

Cylinder (CYLI)



Specific geometric attributes:

Diameter	Diameter of cylinder
Height	Length parallel to Z axis

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Cone (CONE)



Specific geometric attributes:

Dtop	Diameter at top of cone
Dbottom	Diameter at bottom of cone
Height	Length parallel to Z axis

Snout (SNOU)



Specific geometric attributes:

Dtop	Diameter at top of snout
Dbottom	Diameter at bottom of snout
Xoffset	Offset of centre of top from centre of bottom on X axis
Yoffest	Offset of centre of top from centre of bottom on Y axis
Height	Length parallel to Z axis

ⓘ *Only an Xoffset is show in this example, however, both Yoffset and Xoffset may be set.*

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Pyramid (PYRA)



Specific geometric attributes:

Xbottom	Length of bottom of pyramid parallel to X axis
Ybottom	Length of bottom of pyramid parallel to Y axis
Xtop	Length of top of pyramid parallel to X axis
Ytop	Length of top of pyramid parallel to Y axis
Height	Length parallel to Z axis
Xoffset	Offset of centre of top from centre of bottom on X axis
Yoffset	Offset of centre of top from centre of bottom on Y axis

ⓘ *Only a Yoffset is show in this example, however, both Yoffset and Xoffset may be set.*

Circular Torus (CTOR)



Specific geometric attributes:

Rinside	Inside radius in XY plane
Routside	Outside radius in XY plane
Angle	Subtended angle (maximum 180°)

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Rectangular Torus (RTOR)



Specific geometric attributes:

Rinside	Inside radius in XY plane
Routside	Outside radius in XY plane
Height	Length parallel to Z axis
Angle	Subtended angle (maximum 180°)

Dish (DISH)



Specific geometric attributes:

Diameter	Diameter of dish in XY plane.
Height	Height of dish parallel to Z axis
Radius	Knuckle radius

ⓘ *If the knuckle radius is 0 then the dish is represented as a segment of a sphere. If the knuckle radius is greater than 0 then the dish is represented as a partial ellipsoid, generally used to represent a torispherical end to a vessel.*

Copia para EE AA

Sloped Cylinder (SCYL)



Specific geometric attributes:

Diameter	Diameter of sloped cylinder
Height	Length in Z axis from bottom centre to top centre
Xtshear	Inclination of top of cylinder in the XZ axis (in degrees)
Ytshear	Inclination of top of cylinder in the YZ axis (in degrees)
Xbshear	Inclination of bottom of cylinder in the XZ axis (in degrees)
Ybshear	Inclination of top of cylinder in the YZ axis (in degrees)

ⓘ *Only an Xtshear and Ybshear are shown in this example, however, Xtshear, Ytshear, Xbshear and Ybshear may be set in any combination to obtain the required results. The values for these attributes may be +ve or –ve.*

Extrusion (EXTR)



Specific geometric attributes:

Height	Height of extrusion in Z axis

ⓘ *An extrusion is a 2D shape, defined by a series of vertices at each change in direction, extruded through a height. The primitive consists of three element types, i.e. EXTR, LOOP and VERTs.*

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Solid of Revolution (REVO)



180° Rotation 360° Rotation

Specific geometric attributes:

Angle	Rotation angle around X axis (selected rotation line)

ⓘ *A solid of revolution is a 2D shape, defined by a series of vertices at each change in direction, rotated through a specified angle around a specified rotation axis. The primitive consists of three element types, i.e. REVO, LOOP and VERTs.*

Nozzle (NOZZ)



Although a nozzle is classed as a primitive, it is unlike the other primitives in that its geometry is determined in Paragon as part of a catalogue component. Nozzles of different types and geometry may be constructed in Paragon to suit the requirements of the Piping Specification.

The specific nozzle type is referenced from Paragon using the Spref (Specification Reference) attribute.

Specific geometric attributes:

Height	Height between nozzle face and end, i.e. from P1 to P2.

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Appendix B – Example Code

This appendix contains examples of code that provide solutions to each of the exercises. The whole code has not been included in this appendix. Only new/modified code is included so it will require the user to read and understand this code.

Appendix B1 - Example ex2.mac

```
NEW EQUIPMENT /HandWheel

NEW BOX XLEN 100 YLEN 100 ZLEN 100
NEW CYLINDER DIAM 80 HEIG 5
CONN P1 TO P3 OF PREV
NEW BOX XLEN 50 YLEN 50 ZLEN 15
CONN P6 TO P2 OF PREV
NEW DISH DIAM 50 HEIG 15
CONN P2 TO P3 OF PREV

REM ALL
ADD /HandWheel AUTO /HandWheel
```

Appendix B2 - Example ex3.mac

```
$d1=HandWheel
$d2=500

NEW EQUIPMENT /$1

NEW SUBE /$1-Centre
NEW BOX /$1-Centre-Box XLEN 100 YLEN 100 ZLEN 100
NEW CYLINDER DIAM 80 HEIG 5
CONN P1 TO P3 OF PREV
  NEW BOX XLEN 50 YLEN 50 ZLEN 15
CONN P6 TO P2 OF PREV
  NEW DISH DIAM 50 HEIG 15
CONN P2 TO P3 OF PREV

NEW SUBE /$1-Arm-1
  NEW CYLINDER DIAM 50 HEIG ($2/ 2 - 75)
CONN P1 TO P2 OF /$1-Centre-Box
  NEW CTORUS RINS ($2 / 2 - 50) ROUT ($2 / 2)
CONN P0 TO P0 OF /$1-Centre-Box

do !p from 2 to 4
  NEW SUBE /$1-Arm-$!p COPY PREV
    ROTATE BY 90 ABOUT S
enddo

REM ALL
ADD /$1 AUTO /$1
```

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Appendix B3 - Example ex4.mac (fixed)

```
$p

$p HOSE REEL MACRO (VERSION 1.0)

$p -------------------------
$p
$P ENTER REFNO OF THE HOSEREEL (e.g. HOSE-REEL-001)

VAR !NAME |HoseReel|
NEW EQUIP /$!NAME
$P
$P BUILDING HOSEREEL...

VAR !WHEELDIA (500)
VAR !REELDIA (1000)
NEW SUBE /$!NAME-BASE
NEW BOX XLEN 1200 YLEN 725 ZLEN 75

do !N FROM 0 TO 1
  NEW EXTRUSION HEIG 25
    POS W 0 N (-350 + ($!N * 675))  U 0
    ORI Y is E and Z is N
  NEW LOOP
    NEW VERTEX
      POS E 25 S 500 U 0
    NEW VERTEX
      POS E 25 N 500 U 0
    NEW VERTEX
      POS E ($!REELDIA / 2 + 250) N ($!WHEELDIA / 3) U 0
    NEW VERTEX
      POS E ($!REELDIA / 2 + 250 + $!WHEELDIA / 3) N ($!WHEELDIA / 3) U 0
      FRAD ($!WHEELDIA / 3)
    NEW VERTEX
      POS E ($!REELDIA / 2 + 250 + $!WHEELDIA / 3) S ($!WHEELDIA / 3) U 0
      FRAD ($!WHEELDIA / 3)
    NEW VERTEX
      POS E ($!REELDIA / 2 + 250) S ($!WHEELDIA / 3) U 0
enddo
NEW CYLINDER /$!NAME-BASE-AXLE DIAM 100 HEIG 750
  POS E 0 N 0 U ($!REELDIA / 2 + 250)
  ORI Y is E and Z is N
NEW NOZZLE
  POS E 0 N 450 U ($!REELDIA / 2 + 250)
  ORI Y is E and Z is U
  HEIG 100
  CATR /DICHTFL_C/DEZFBR0NN

NEW SUBE /$!NAME-HOSE
  POS E 0 N 0 U ($!REELDIA / 2 + 250)
  ORI Y is D and Z is S
  do !N FROM 0 to 1
    NEW CYLINDER DIAM $!REELDIA HEIG 10
      POS E 0 N 0 U (-305 + ($!N * 610))
  enddo
  do !I FROM 0 TO 5
    do !J FROM 0 TO 1
      NEW CTORUS RINS (($!REELDIA / 2) - 150) ROUT (($!REELDIA / 2) - 50) ANGL
180
        POS E 0 N 0 D (-250 + ($!I * 100))
        ROTATE BY ($!J * 180) ABOUT S WRT /*
    enddo
  enddo
```

Fix1: READ syntax replaced with standard variable declaration

Fix2: TO instead of TOO

Fix3: FRAD instead of PRAD

Fix4: S instead of S10E

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

```
  NEW CYLINDER DIAM 100 HEIG ($!REELDIA / 3)
    CONN P1 TO P2 OF PREV
  NEW CYLINDER DIAM 140 HEIG 25
    CONN P1 TO P2 OF PREV
  NEW CYLINDER DIAM 110 HEIG 50
    CONN P1 TO P2 OF PREV
  NEW CYLINDER DIAM 140 HEIG 25
    CONN P1 TO P2 OF PREV
  NEW CONE DTOP 110 DBOT 80 HEIG 200
    CONN P1 TO P2 OF PREV
  NEW REVOLUTION ANGL 360
    NEW LOOP
      NEW VERTEX
      NEW VERTEX
        POS W 10 N 0 U 0
      NEW VERTEX
        POS W 10 N 45 U 0
        FRAD 5
      NEW VERTEX
        POS W 0 N 45 U 0
        FRAD 5
  REVO
    CONN P1 TO P2 OF PREV
    ROTATE BY 90 ABOUT D

NEW SUBE /$!NAME-CENTRE
  NEW BOX /$!NAME-CENTRE-BOX XLEN 100 YLEN 100 ZLEN 100
    CONN P6 TO P2 OF /$!NAME-BASE-AXLE
  NEW CYLINDER DIAM 80 HEIG 5
    CONN P1 TO P3 OF PREV
  NEW BOX XLEN 50 YLEN 50 ZLEN 15
    CONN P6 TO P2 OF PREV
  NEW DISH DIAM 50 HEIG 15
    CONN P2 TO P3 OF PREV

NEW SUBE /$!NAME-ARM-1
  VAR !POS POS /$!NAME-CENTRE-BOX
  POS $!POS
  NEW CYLINDER DIAM 50 HEIG ($!WHEELDIA / 2 - 75)
    CONN P1 TO P2 OF /$!NAME-CENTRE-BOX
  NEW CTORUS RINS ($!WHEELDIA / 2 - 50) ROUT ($!WHEELDIA / 2)
    CONN P0 TO P0 OF /$!NAME-CENTRE-BOX

DO !P FROM 2 TO 4
  NEW SUBE /$!NAME-ARM-$!P COPY PREV
    ROTATE BY 90 ABOUT S
enddo

REM ALL
ADD /$!NAME AUTO /$!NAME
EQUIP
$P
$P DONE!
$P
```

Fix5: P1 to P2 instead of P1 to P1

Fix6: BOX instead of BOXES

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Appendix B4 - Example ex4.pmlfnc

```
define function !!ex4()

NEW EQUIP /HoseWheel

VAR !WHEELDIA (500)
VAR !REELDIA (1000)

NEW SUBE /$!NAME-BASE
--------------------
--
--  As original Macro
--
--------------------
EQUIP

endfunction
```

Middle part, same as the original macro

Appendix B5 - Example hosewheel.pmlfnc

```
define function !!hosewheel(!name is STRING, !wheelDia is REAL)

NEW EQUIP /$!NAME

VAR !REELDIA (1000)

NEW SUBE /$!NAME-BASE
--------------------
--
--  As original Macro
--
--------------------
EQUIP

endfunction
```

Middle part, same as the original macro

Appendix B6 - Example highlighttype.pmlfnc

```
define function !!highlightType(!type is STRING, !col is REAL) is ARRAY

  !collection = object COLLECTION()
  !collection.type(!type)
  !collection.scope(!!ce)
  !results = !collection.results()

  do !element values !results
    ENHANCE $!element col $!col
  enddo

  return !results

endfunction
```

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Appendix C – Supporting PML Toolbar

This appendix explains the PML Toolbar, supplied to support the development of PML forms and the training course. All the PML for this utility will be found in the supplied files, within the Utilities folder.

Appendix C.1 - The PML Toolbar

The following toolbar provides all the supplied functionality:



Toolbar Gadget	Description
	Runs the command “PML REHASH”
	Runs the command “PML REHASH ALL”
	The available methods of finding available PML forms
	The available forms to be reloaded, viewed or opened
	Runs the command “PML RELOAD FORM” for the selected form
	Runs the command “SHOW” for the selected form
	Opens the selected form in the default Windows program
	Opens the E3D User Manuals index
	Show the Available E3D Colours form (see Appendix C.2)
	Show the Available E3D Icons form (see Appendix C.3)
	Show the Training Examples form (see Appendix C.4)

Appendix C.2 - Available E3D Colours Form

This form will display the available colours within the E3D session. Dynamically built, it will always be correct for the current session.

If the preview form is shown, a larger version of the coloured button is available along with more information about the colour. To update the colour preview, choose a different colour from the main form.

To get a record of the colours, they can be saved to an Excel spreadsheet. This will record the colour name, name and if it’s a mix colour the RGB values.

The purpose of this form is to aid with setting colours (e.g. highlighting, gadgets etc) within any customisation

Copia non AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.

Appendix C.3 - Available E3D Icons Form



This form will search the chosen PMLLIB search path for .png image files and display them in the grid control gadget. A preview of the image will be displayed and it will be possible to gain a larger preview from the context menu

As the PMLLIB search paths could be large and contain many images, it is possible to only search specific sub folders. Standard grid control functionality is also available to help filter the search results.

The purpose of the form is to visualise all the images available in the PMLLIB search path. As the images are within the search path, these images are therefore available for use in any customisation.

ⓘ *Depending on the size of the pml.index file, this form can take a while to collect and display all the information.*

Appendix C.4 – Training Examples Form



This form will provide quick access to all the training examples supplied as part of this course. Split into types, each example can be shown inside E3D or opened in the default Windows Program.

The form builds itself based on the contents of a .xls file saved within the same folder as the form definition.

By making choices in the top two sections, different examples will be available at the bottom of the form.

The examples can be displayed within E3D or their definition opening in the default Windows program. Choosing between the icons on the right will switch this mode:



In "Run" Mode



In "Open" mode

The purpose of the form is to speed up access to the training examples and to prompt their future use as reference.

ⓘ *As you attend further courses, more of the options across the top will become available*

Copia para EE AA

© Copyright 1974 to current year.
AVEVA Solutions Limited and its subsidiaries.
All rights reserved.